
COUNTIFS with multiple criteria and OR logic

Generic formula 

= SUM(COUNTIFS(range,{"red","blue","green"}))

Summary 

To count with multiple criteria and OR logic, you can use the COUNTIFS function with an array
constant. In the example shown, the formula in H6 is:

= SUM(COUNTIFS(D4:D11,{"complete","pending"}))

Explanation 

By default, the COUNTIFS function applies AND logic. When you supply multiple conditions, all
conditions must match in order to generate a count.

One solution is to supply multiple criteria in an array constant like this:

{"complete","pending"}

This will cause COUNTIFS to return two results: a count for "complete" and a count for "pending",
packaged in an array result like this:

{4,2}

To get a final total, we wrap COUNTIFS inside SUM. The SUM function then sums all items in the
array and returns the result.

Adding another OR criteria

You can add one additional criteria to this formula, but you'll need to use a single column array for
one criteria and a single row array for the other. So, for example, to count orders that are
"Complete" or "Pending", for either "Andy Garcia" or "Bob Jones", you can use:

= SUM(COUNTIFS(D4:D11,{"complete","pending"},C4:C11,{"Bob 
Jones";"Andy Garcia"}))

Note we use a semi-colon for the second array constant, which creates a vertical array. This works
because Excel "pairs" elements in the two array constants, and returns a two dimensional array of
results inside SUM like this:

= SUM({1,0;1,0}) 

If you have more than two criteria, consider a SUMPRODUCT formula as described here.

Cell reference for criteria

To use a cell reference for criteria, you can use an array formula like this:

= {SUM(COUNTIFS(range1,range2))}

Where range1 is the criteria range, and range2 contains criteria.

Wildcards and double-counting

COUNTIF and COUNTIFS support wildcards, but you need to be careful not to double count when
you have multiple "contains" conditions with OR logic. See this example for more information

Related formulas 

Count cells equal to either x or y
COUNTIF counts the number of cells in a range that match
supplied criteria. In this case, the criteria for the first COUNTIF is
"apples" and the criteria for the second COUNTIF is "pears". The
first COUNTIF returns the count of cells in B5:B10 equal...

Count cells not equal to x or y
The COUNTIFS function counts the number of cells in a range
that meet one or more supplied criteria. All conditions must pass
in order for a cell to be counted. In the example shown, there is a
list of colors in column B in a named range called rng...

Count cells that contain either x or y
When you count cells with "OR logic", you need to be careful not
to double count. For example, if you are counting cells that
contain "abc" or "def", you can't just add together two COUNTIF
functions, because you may double count cells that contain...

SUMPRODUCT count multiple OR criteria
Working from the inside out, each criteria is applied with a
separate ISNUMBER + MATCH construction. To generate a count
of rows in column one where the value is A or B we use:
ISNUMBER ( MATCH ( B5:B11 , { "A" , "B" } , 0 )...

Count matching values in matching columns
In this example, the goal is to count "z" or "c" values in the named
range data , but only when the column header is "A" or "B". The
formula used to perform this calculation is based on the
SUMPRODUCT function : = SUMPRODUCT ( ISNUMBER ( MATCH
(...

Related functions 

Excel COUNTIFS Function
The Excel COUNTIFS function returns the count of cells that
meet one or more criteria. COUNTIFS can be used with criteria
based on dates, numbers, text, and other conditions. COUNTIFS
supports logical operators (>,...

Related videos 

How to use the COUNTIFS function
In this video we'll look at how to use the COUNTIFS function to
count cells that meet multiple criteria in a set of data.   

Good links 

Multiple criteria with one or two ranges (Excelxor)

